
Hoofdstuk 8.

Zoeken.

Bepaalde problemen komen voor in heel veel verschillende computertoepassingen. Het is dus belangrijk goede oplossingen voor deze problemen te kennen. Het kan ook praktisch zijn de oplossingen van deze problemen op voorhand te programmeren zodanig dat men ze als componenten kan aanwenden wanneer men ze nodig heeft. In dit hoofdstuk zullen, bij wijze van voorbeeld, enkele zoekalgoritmen besproken worden. Het betreft het zoeken naar een bepaalde waarde in een Array en het zoeken naar een bepaalde string in een tekst.

8.1. Zoeken naar een bepaalde waarde in een Array.

De algoritmen die hier zullen behandeld worden zoeken in een Array of Records. Elk Record bevat een specifiek veld, Key. Het algoritme zoekt het Record waarvan het Key veld een gegeven waarde heeft. Hier zullen we aannemen dat alle elementen in de Array benut worden, met uitzondering van het eerste, waarvan de index de waarde 0 heeft. Dergelijke datastructuur is beschreven in tabel 8.1.

Het type van het Key veld moet relationele operaties toelaten. Indien dat niet het geval is, bijvoorbeeld omdat het een string bevat, moet de operator = vervangen worden door een booleaanse procedure die hetzelfde effect heeft maar toepasbaar is op het type van Key.

TYPE

  IndexType = [0 .. TableSize];

  Item      = RECORD
                Key : (* Any type that accepts the = and # 

                         operators *);

                (* other fields *)

              END (* Item *)

  Table     = Array IndexType OF Item;



Tabel 8.1. Data structure for Array search algorithms.

Deze structuur zou bv. kunnen gebruikt worden voor een telefoonlijst. Elke Item zou dan een naam en een telefoonnummer bevatten. Normale telefoonlijsten zijn alfabetisch gerangschikt naar naam; daardoor zal het zoeken naar een record waarvan het naamveld gegeven is veel sneller zijn dan het zoeken naar een record waarvan het nummerveld als sleutel gebruikt wordt.

De verschillende zoek algoritmen zullen opgenomen worden in een cardinale functie procedure, Search. Deze procedure heeft twee parameters, de eerste is een open array waarin moet gezocht worden, de tweede de gezochte waarde. Indien de gezochte waarde gevonden wordt, krijgt Search de waarde van de overeenstemmende index, zoniet de waarde 0.

Tabel 8.1 toont de nuttige delen van de data definities en een voorbeeld van gebruik van de procedure Search.

8.1.1. Voor de hand liggend algoritme.

Tabel 8.2 geeft een voor de hand liggende versie van de procedure Search: vertrekkend van het laatste element van de array, worden de key velden van alle opeenvolgende records met de gezochte waarde vergeleken, tot de waarde gevonden is of tot het nulde element van de array bereikt wordt. De standaard functie HIGH wordt gebruikt om de afmeting van de actuele array parameter te vinden.

...

CONST
  SIZE = ...; (*Number of records in array *)

  ... 

TYPE
  KeyType = ...;

  (* any type accepting relational operators *)

  ArrayType  = ARRAY[0..Size] OF RecordType;

  RecordType = RECORD
                 Key : KeyType;

                 ...

               END;

  ...

VAR
  Data : ArrayType;

  Key  : KeyType;

  Found: [0..Size];

  ...

BEGIN
  ...

  Key := ...;

  Found := Search(Data,Key);

  IF Found = 0

    THEN WrStr(“Item not found”)

    ELSE (* Write contents of Data[Found] *)

      ...

  END (* IF *);

...    

Table 8.1. Gebruik van de Search procedures.

PROCEDURE Search (VAR A: ARRAY OF RecordType,

                    Key: KeyType)

                 : CARDINAL;

(* Straight Search *)

  VAR
    i : CARDINAL;

BEGIN
  i := HIGH(A);

  WHILE A[i].Key # Key AND i # 0 DO DEC(i) END;

  RETURN i

END Search;

Table 8.2. Voor de hand liggende Search procedure.

De zoektijd hangt af van de plaats waar het element gevonden wordt. Indien het element niet gevonden wordt is de zoektijd maximaal. Deze tijd is evenredig met het aantal elementen in de array, vermits er, per element, twee relationele expressies en een booleaanse expressie moeten geëvalueerd worden.

8.1.2. Zoeken met een sentinel.

Tabel 8.3 geeft een betere versie van de procedure Search. In de eerste versie moesten twee relationele expressies en een booleaanse expressie geëvalueerd worden per element. Dit komt omdat men niet op voorhand weet of het gezochte element al of niet aanwezig is en men dus telkens moet nagaan of het einde van de array nog niet bereikt is. Dit kan vermeden worden door eerst zelf het gezochte element in het nulde element van de arrray te plaatsen. Vermits er van achter naar voor gezocht wordt, betekent het vinden van het zelf geplaatste element de afwezigheid van het gezochte wat wel degelijk overeenkomt met de waarde 0 van de index.

PROCEDURE Search (VAR A: ARRAY OF RecordType,

                    Key: KeyType)

                 : CARDINAL;

(* Sentinel Search *)

  VAR
    i : CARDINAL;

BEGIN
  i := HIGH(A);

  A[0].Key := Key;

  WHILE A[i].Key # Key DO DEC(i) END;

  RETURN I

END Search;

Table 1.3. Sentinel Search procedure.

8.1.4. Recursief zoeken in een geordende tabel

Zoals reeds vermeld in hoodstuk 7, bestaat de efficiëntste manier om een naam in een telefoongids op te zoeken (alfabetisch gesorteerde lijst) erin de middelste naam te nemen, deze te vergelijken met de op te zoeken naam en, naargelang het resultaat van de vergelijking, hetzij de eerste, hetzij de tweede helft te verwijderen. Zolang de overblijvende lijst meer dan een naam bevat, wordt dezelfde procedure toegepast op de overblijvende lijst. 

Indien de oorspronkelijke lijst n elementen bevatte, zal de overblijvende lijst na k vergelijkingen nog n/2k elementen bevatten. Wanneer deze overblijvende lijst nog maar één element bevat is de zoektocht beëindigd. Het totale aantal vergelijkingen is dus (log2n, waarbij het ( symbool gebruikt wordt om “het eerste gehele getal groter dan” aan te geven.

Dit algoritme, dat gekend staat onder de naam “Binary Search” is voorgesteld in figuur 8.1. terwijl tabel 8.4 de overeenstemmende versie van de procedure Search geeft. Het is wel belangrijk te onthouden dat dit algoritme enkel werkt indien de tabel stijgend geordend is naar de waarden van het Key veld.

Fig.8.1. Het recursief binair zoek algoritme.

PROCEDURE Search (VAR A: ARRAY OF RecordType,

                  Key: KeyType)

                 : CARDINAL;

(* Recursive binary Search *)

  VAR
   Min,Max,m: CARDINAL;

  PROCEDURE src(Min,Max: CARDINAL);

   BEGIN
      m := (Min+Max) DIV 2;

      IF Min # Max  THEN
          IF a[m].Key >= Key

            THEN src(Min,m)

            ELSE src(m+1,Max)

          END;

      END;

   END src;

BEGIN
  Min := 1; Max := HIGH(A);

  src(Min,Max)

  IF a[m].Key = Key

   THEN RETURN m

   ELSE RETURN 0

  END
END Search;

Table 8.4. Recursive binary search procedure.

Daar procedure oproepen instructies zijn die vrij veel tijd en geheugenruimte kosten is het nuttig op te merken dat, vermits de procedure src geen lokale variabelen heeft en daarbij bijzonder eenvoudig is, de recursieve oproepen van die procedure zonder meer kunnen gesimuleerd worden door middel van een WHILE lus. Tabel 8.5 toont de overeenstemmende versie van de procedure Search. De verschillen tussen deze twee versies van Search zijn aangeduid door het gebruik van schuine lettertypes. In feite doen beide procedures strikt hetzelfde: de opeenvolgende halveringen van het zoekdomein resulteren in de eerste van de recursieve oproepen van src, in de tweede van de WHILE lus.

PROCEDURE Search (VAR A: ARRAY OF RecordType,

                  Key: KeyType)

                 : CARDINAL;

(* Iterative binary Search *)

  VAR
   Min,Max,m: CARDINAL;

BEGIN
  Min := 1; Max := HIGH(A);

  WHILE Min < Max DO

    m := (Min+Max) DIV 2;  

    IF a[m].Key >= Key

      THEN Max := m

      ELSE Min := m+1

    END; (*IF*)

  END; (*WHILE*)

  IF a[m].Key = Key

    THEN RETURN m

    ELSE RETURN 0

  END; (*IF*)

END Search;



Table 8.5. Iterative binary search procedure.

8.1.5. Performantie beschouwingen.

In beide algoritmen voor ongeordende tabellen is de zoektijd evenredig met de lengte van de tabel. De sentinel methode is echter sneller, want het aantal vergelijkingen dat moet gebeuren is gehalveerd ten opzichte van de voor de hand liggende methode. Bij beide algoritmen zal de zoektijd maximaal zijn wanneer het gezochte element niet aanwezig is of helemaal vooraan in de tabel staat.

Daar het “Binary Search” algoritme recursief de tabel halveert tot de overblijvende tabel nog maar één enkel element bevat wordt de zoektijd niet beïnvloed door uitkomst van het zoeken. Er werd hogerop aangetoond dat deze zoektijd evenredig is met het logaritme in basis 2 van het aantal elementen.

Tabel 8.6 geeft, voor een tabel met 10000 elementen de zoektijden die gemeten werden op een Compaq 400CX laptop computer met Top Speed Modula. De programma’s die aangewend werden om deze metingen te verrichten vindt men in appendix 8.A. In deze programma’s kan men opmerken dat, om preciese tijdsmetingen te verrichten ondanks de 10 mS resolutie van de DOS klok, de tijdsmeting gaat over 1000 hehalingen van dezelfde zoekactie. Tengevolge de vele nevenactiviteiten van een modern besturingssysteem zoals Windows, wordt het aanbevolen dergelijke metingen te doen met enkel een vrij primitief besturingssysteem zoals MS/DOS.

First key
6100
5050
33
11

Middle key
3020
2580
33
11

Last key
<10
<10
33
11

Absent key
6150
5110
33
11

Table 8.6. Search times for various search algorithms.

8.2 Zoeken naar een bepaalde string in een tekst.

8.2.1. Inleiding.

Algoritmen om een bepaalde string (dat is een verzameling opeenvolgende tekens) in een tekst te zoeken worden in vele programma’s aangewend. Men vindt ze in tekstverwerkers, in spellingscorrectieprogramma’s, in compilatoren en ook in software bestemd voor andere takken van de wetenschap, zoals bv. de biologie waar men deze algoritmen gebruikt om specifieke sequenties van basen in DNA te zoeken.

In dit hoofdstuk worden zowel de tekst als de string opgeslagen in arrays met characters.

Bij de voorbeeld programma’s gebruikt men twee ad-hoc types, het type StringType en het type TextType gedefiniëerd in tabel 8.7. Zoals in de andere zoekalgoritmen wordt het eerste element van de arrays voorbehouden voor hulpvariabelen.

CONST

   MaxStringLength = 100;

   MaxTextLength = 20000;

TYPE

   StringType = RECORD

                  Lng : [0..MaxStringLength];

                  Txt : ARRAY [0..MaxStringLength] OF CHAR;

                END;

   TextType   = RECORD

                  Lng : [0..MaxTextLength];

                  Txt : ARRAY [0..MaxTextLength] OF CHAR;

                END;

Table 8.7. Data declarations for string search programs.

De besproken algoritmen zoeken vanaf het begin van de tekst tot zij het eerste voorkomen van de string of het einde van de tekst bereiken. Ze uitbreiden om alle voorkomens op te sporen is triviaal en wordt aan de lezer als oefening gesuggereerd.

Drie verschillende algoritmen zullen besproken worden: 

· het “Brute Force” (BF) algoritme is het eenvoudigst maar meestal ook het traagst.

· het Knuth, Morris en Pratt (KMP) algoritme, dat in 1974 uitgevonden werd, is meestal veel sneller dan het BF algoritme.

· het Boyer en Moore (BM) algoritme, dat in 1977 uitgevonden werd is, statistisch gezien, het meest performante, maar het verschil met KMP is zelden opvallend.

8.2.2. Het “Brute Force” algoritme.

Het algoritme werkt principieel als volgt: de string wordt verschoven langs de tekst tot men een positie vindt waarbij het eerste karakter van de string gelijk is aan een karakter van de tekst. Dan worden de volgende karakters van de string vergeleken met de overeenstemmende karakters van de tekst. Indien alle karakters gelijk blijken te zijn is de string gevonden, zoniet schuift men de string verder langs de tekst.

Figuur 8.2 stelt het BF algoritme voor. Tabel 8.8 bevat een eenvoudige functie procedure Search gebaseerd op het BF algoritme. De procedure heeft twee parameters: de tekst en de string.

De functie bepaalt de index in de array text waar de string gevonden is. Indien de string niet gevonden wordt keert de functie een waarde terug die groter is dan de lengte van de text verminderd met de lengte van de string.

PROCEDURE Search(VAR Text: TextType; VAR String: StringType)

  VAR j, jmax : CARDINAL;

BEGIN
  j := 1; jmax := Text.Lng – String.Lng;

  REPEAT
    WHILE (Text.Txt[j]#String.Txt[1]) AND (j<=jmax) 

      DO j := j+1 END;

    IF j <= jmax THEN
      i := 1;

      WHILE (Text.Txt[j+i-1]=String.Txt[i]) AND (i<String.Lng) 

        DO i := i+1 END;

    END; (* IF *)

    j := j + 1;

  UNTIL (i = String.Lng) OR (j > jmax);

  RETURN j-1

END Search;

Table 8.8. A straightforward implementation of the BF algorithm.

De aandachtige lezer zal vaststellen dat beide WHILE lussen een dubelle test bevatten, waarvan er een dient om na te gaan of het einde van de array nog niet bereikt werd. Een niet te verwaarlozen performantiewinst kan bekomen worden indien men hier sentinel technieken gaat toepassen. Voor de eerste WHILE lus volstaat het de gezochte string achter de tekst toe te voegen, en voor de tweede WHILE lus volstaat het een karakter dat zeker niet in de tekst kan voorkomen achteraan de string toe te voegen. Tabel 8.9 toont een alzo verbeterde versie van Search.

PROCEDURE Search(VAR Text: TextType; VAR String: StringType)

  VAR j, jmax : CARDINAL;

BEGIN
  WITH Text DO (* sentinel for the Text loop *)
    Txt[Lng+1] := String.Txt[1];
  END;

  WITH String DO  (* sentinel for the String loop *)

    Txt[Lng+1] := CHR(0); 

  END;
  j := 1; jmax := Text.Lng – String.Lng;

  REPEAT
    WHILE (Text.Txt[j]#String.Txt[1]) DO j := j+1 END;

    IF j <= jmax THEN
      i := 1;

      WHILE (Text.Txt[j+i-1]=String.Txt[i]) DO i := i+1 END;

    END; (* IF *)

    j := j + 1;

  UNTIL (i > String.Lng) OR (j > jmax);

  RETURN j-1

END Search;

Table 8.9. A sentinel implementation of the BF search algorithm.
8.2.3. Het KMP algoritme.

8.2.3.1. Het onderliggende idee.

Wanneer, in de tweede WHILE lus van het BF algoritme, een verschil vastgesteld wordt, keert men zonder meer terug naar de eerste WHILE lus en schuift men de string stapsgewijs verder langs de tekst. Informatie over de volgende karakters in de tekst die men had kunnen opdoen tijdens de uitvoering van de tweede WHILE lus gaat alzo verloren. Hoe men deze informatie nuttig zou kunnen gebruiken wordt getoont in volgend voorbeeld. Het voorbeeld toont aan hoe de string langs de tekst geschoven wordt.

b a b c b a b c a b c a a b c a b c a b c a c a b c a c

a b c a b c a c a b

(
Initiëel zijn het begin van de tekst en van de string links gealigneerd en wordt er reeds een verschil (aangeduid door () vastgesteld bij het eerste karakter van de tekst. Dit verschil geeft aan dat de string, minstens één positie naar rechts dient opgeschoven te worden.

b a b c b a b c a b c a a b c a b c a b c a c a b c a c

  a b c a b c a c a b

        (
Nu kan de vergelijking verder gaan tot het vijfde karakter van de tekst alvorens een verschil vastgesteld wordt. Deze vaststelling toont niet alleen aan dat positie 2 in de tekst geen goed startpunt is, maar ook dat de string meteen 4 posities naar rechts kan verschoven worden, vermits karakters 3, 4 en 5 onmogelijk gelijk kunnen zijn aan de “a” waarmee de string begint.
b a b c b a b c a b c a a b c a b c a b c a c a b c a c

          a b c a b c a c a b

                        (
Wanneer men verder gaat met het vergelijken van karakters vanaf positie 6 in de tekst, zal men het eerste verschil vinden in positie 13. Als men dan de string en de plaats van het verschil onderzoekt kan men besluiten dat voor de volgende poging de string kan opgeschoven worden tot in positie 9 van de tekst, maar dat het niet nodig is de gelijkheid van de karakters in posities 9 tot en met 12 te verifiëren. 
b a b c b a b c a b c a a b c a b c a b c a c a b c a c

                a b c a b c a c a b

                        (
In positie 13 blijkt er nog steeds een verschil te bestaan, maar hieruit kan men nu wel afleiden dat de volgende poging zal kunnen beginnen met het begin van de string in positie 13.

b a b c b a b c a b c a a b c a b c a b c a c a b c a c

                        a b c a b c a c a b

                                      (
Deze poging brengt een verschil in positie 20 aan het licht en laat toe de string vooruit te schuiven tot in positie 16. Verdere vergelijkingen hoeven maar te gebeuren vanaf positie 20. Deze maal, uiteindelijk, kunnen de vergelijkingen verder gaan tot het einde van de string, zonder dat er een verschil vastgesteld wordt.

b a b c b a b c a b c a a b c a b c a b c a c a b c a c

                              a b c a b c a c a b

                              ^                   (
Dit voorbeeld toont dat een string kan gezocht worden in een tekst zonder ooit achteruit te gaan bij het vergelijken van karakters (soms moet hetzelfde karakter wel opnieuw vergeleken worden). Dit algoritme vereist dat de string op voorhand ontleed is geweest om te bepalen hoeveel posities naar voren mag geschoven worden wanneer een verschil op een bepaalde plaats vastgesteld wordt.

8.2.3.2. Het zoekalgoritme.

Om de karakters die moeten vergeleken worden aan te duiden in de string en in de tekst zal men resp. de indices i en j gebruiken. De string t.o.v. de tekst verschuiven komt er dan op neer de waarden van i en/of j te wijzigen. De eenvoudigste wijze om het algoritme te implementeren bestaat erin gebruik te maken van een tabel Next[i] waarin, voor elke waarde van i de volgende waarde gegeven wordt die i moet aannemen in geval een verschil vastgesteld wordt in positie i. Tabel 8.10 geeft de inhoud van de tabel Next die oveenkomt met de string die in het voorbeeld behandeld werd. Wanneer een verschil gevonden wordt in positie i wordt de string i-Next[i] plaatsen vooruit geschoven ten opzichte van de tekst. Een waarde 0 in Next betekent dat de string kan opgeschoven worden tot aan de plaats die op het vastgestelde verschil volgt. Het algoritme om de Next tabel op te bouwen wordt besproken in paragraaf 8.2.3.3.

i =
1  2  3  4  5  6  7  8  9 10

String.Txt[i] =

Next[i] =

Shift[i] =
a  b  c  a  b  c  a  c  a  b

0  1  1  0  1  1  0  5  0  1

1  1  2  4  4  5  7  3  9  9

Table 8.10. The next table for a given string.

Het eigenlijke zoekalgoritme vindt men in tabel 8.11. Het bevat twee geneste WHILE lussen, de buitenste schuift hoofdzakelijk de i en j indexen vooruit, terwijl de binnenste overeenstemmende karakters vergelijkt en de i index aanpast telkens een verschil vastgesteld wordt.

j := 1; i := 1;
WHILE (i <= String.Lng) AND (j <= Text.Lng) DO
  WHILE (i>0) AND (Text.Txt[j] # String.Txt[i]) DO i := Next[i] END;

  i := i + 1; j := j + 1;

END; (* WHILE *)

Table 8.11. The core of the KMP search algorithm

8.2.3.3. Het opbouwen van de Next tabel.
Wanneer in een KMP opzoeking een verschil tussen Text.Txt[j] en String.Txt[i] vastgesteld wordt zijn volgende asserties waar:
Text.Txt[j-i+1] = String.Txt[1]

...

Text.Txt[j-1]   = String.Txt[i-1]

Text.Txt[j]     # String.Txt[i]

Na de vaststelling van het verschil , wordt de string vooruit geschoven tot wanneer, opnieuw alle karakterparen die Text.Txt[j] voorafgaan gelijk zijn, of tot wanneer String.Txt[1] de positie van Text.Txt[j+1] heeft bereikt. In de veronderstelling dat een nieuwe overeenstemming gevonden werd, en dat String.Txt[n] tegenover Text.Txt[j] komt te staan, dan zijn volgende asserties waar:

Text.Txt[j-n+1] = String.Txt[1]

...

Text.Txt[j-1]   = String.Txt[n-1]

Door de twee verzamelingen asserties te combineren kan men er een bekomen waarin geen referenties naar Text meer voorkomen.

String.Txt[i-n+1] = String.Txt[1]

...

String.Txt[i-1]   = String.Txt[n-1]

Deze asserties betekenen dat de patroon gevormd door de eerste (n-1) karakters van de string opnieuw voorkomt in posities (i-n+1) to (i-1). Het bepalen, voor een gegeven waarde van i, van de overeenstemmende waarde van n kan dus gebeuren door het bepalen van de langste patroon, korter dan i karakters, die twee maal in de string voorkomt, één maal startend in positie (1), de andere maal eindigend in positie (i-1). 

Tabel 8.12 toont, voor de string die in het voorbeeld gebruikt werd, de waarde van n[i] die met elke waarde van i overeenkomt. Conventioneel heeft men de waarde 0 gegeven aan n[1] en de waarde 1 voor alle waarden van i voor dewelke geen verschuifbaar patroon bestaat. De waarde 1 hiervoor is coherent met het feit dat gelijk welke patroon van lengte 0 gelijk is aan gelijk welke andere patroon van lengte 0.

i =
1  2  3  4  5  6  7  8  9 10
 n[i] =

1

2

3

4

5

6

7

8

9

10
a  b  c  a  b  c  a  c  a  b

^

      ^

         a  b  c  ^

                        ^

                        a  ^
0

1

1

1

2

3

4

5

1
2

Table 8.12. The value of n for different values of i.

Een algoritme om n[k+1], vertrekkend van n[k] te vinden zal in de volgende paragraaf ontwikkeld worden. Vermits n[1] per definitie de waaarde 0 heeft volstaat het aangekondigde algoritme om de volledige verzameling van n[i] waarden te bepalen.
Eerst en vooral dient men op te merken dat, telkens String.Txt[k] = String.Txt[n[k]],

n[k+1] = n[k] + 1. Dit is het geval in bovenstaand voorbeeld voor k= 4,5,6,7,9 en 10)

Nu zal er aangetoond worden dat, wanneer String.Txt[k] ( String.Txt[n[k]], het algoritme dat reeds beschreven werd in paragraaf 8.2.3.2 aangewend kan worden om n[k+1] te vinden.

Zoals eerder vermeld, bevat de Next tabel, voor elke positie i in de string waar een verschil kan vastgesteld worden, de index van het volgende karakter in de string dat moet tegenover karakter j van de tekst gebracht worden. Een 0 in de tablel Next betekent dat de volledige string kan verschoven worden tot na de plaats in de tekst waar het verschil vastgesteld werd en dat de volgende vergelijking zal gebeuren tussen String.Txt[1] en Text.Txt[j+1]

Dit impliceert dat Next[1] per definitie gelijk is aan 0.

Vertrekkend met n[1] en Next[1] kunnen nu alle volgende waarden in de vectoren n en Next berekend worden. Tabel 8.13 toont een procedure die n[k+1] berekent vertrekkend van de waarden van n[k] en Next[1] tot Next[k-1]. Deze procedure werd reeds beschreven in paragraaf 8.2.3.2. maar wordt hier toegepast op de string en op een deel van dezelfde string.
t := n[k];

WHILE t>0 AND String[k]#String[t] DO
  t := Next[t]

END; (* WHILE *)

n[k+1] := t+1;

Table 8.13. Computation of n[k+1] after n[k] has been computed.

Om dit algoritme goed te verstaan kan het nuttig zijn het met de hand toe te passen op het voorbeeld van tabel 8.12 en, bijvoorbeeld, n[9] te berekenen nadat n[8] = 5 reeds was gevonden (de waarden van Next kunnen in tabel 8.10 gevonden worden).

Voor de waarde i = 8 had men gevonden dat 4 opeenvolgende karakters gelijk waren wanneer String.Txt[5] gealigneerd werd met String.Txt[8]. Hieruit, en met behulp van het algoritme beschreven in tabel 8.13, kan men afleiden dat t de waarde van n[8] krijgt, het is te zeggen 5.

Dit komt overeen met volgende relatieve posities van de twee versies van de string

1  2  3  4  5  6  7  8  9 10
a  b  c  a  b  c  a  c  a  b

         a  b  c  a  b  c  a  c  a  b

                     ^

Vermits String.Txt[8] ( String.Txt[5], krijgt t de waarde van Next[5] = 1;

1  2  3  4  5  6  7  8  9 10
a  b  c  a  b  c  a  c  a  b

                     a  b  c  a  b  c  a  c  a  b

                     ^

Vermits nu String.Txt[8] ( String.Txt[1], krijgt t de waarde van Next[1] = 0;
1  2  3  4  5  6  7  8  9 10
a  b  c  a  b  c  a  c  a  b

                     a  b  c  a  b  c  a  c  a  b

                     ^

Daar t nu niet meer >0 is, zal de WHILE lus beëindigd worden, en vermits er geen mogelijke overeenstemming gevonden werd krijgt n[9] de waarde 1.

De waarde van Next[k] berekenen, nadat de waarde van n[k] gevonden werd, is vrij eenvoudig: Wanneer een verschil ontdekt wordt tussen element String.Txt[i] en element Text.Txt[j], wordt, conceptueel, de string vooruit geschoven zodanig dat positie String.Txt[n[i]] tegenover element Text.Txt[j] komt te staan en deze twee elementen kunnen vergeleken worden.

Indien String.Txt[n[i]] = String.Txt[i] zal deze vergelijking op evidente wijze een verschil aantonen en kan de string verder vooruitgeschoven worden, in overeenstemming met Next[n[i]].

Indien, integendeel, String.Txt[n[i]] ( String.Txt[i] dan is de uitslag van de vergelijking tussen String.Txt[n[i]] en Text.Txt[j] onvoorspelbaar en moet het vergelijkingsproces karakter per karakter verder gaan tot een verschil ontdekt wordt.

Hieruit kan men besluiten dat Next[k] eenvoudigweg kan bepaald worden door de instructie die weergegeven is in tabel 8.14.

IF String[k] # String[n[k]] THEN Next[k] = n[k]

                            ELSE Next[k] = Next[n[k]] 

Table 8.14. Computation of Next[k] in function of n[k].

Door het algoritme om n[k] te berekenen (Tabel 1.13) te combineren met het algoritme om Next[k] te berekenen (Tabel 1.14) kan men een algoritme opbouwen dat Next[k] berekent zonder expliciet langs n[k] om te gaan. Dit algoritme wordt verwezenlijkt door het deelprogramma weergegeven in tabel 8.15.
i := 1; k := 0; Next[1] := 0;

WHILE I < SLength DO
  WHILE (k > 0) AND (String[i]#String[k]) DO k := Next[k] END;

  k := k + 1; i := i + 1;

  IF String[i] = String[k]

    THEN Next[i] := Next[k]

    ELSE Next[i] := k

  END; (* IF *)

END; (* WHILE *)

Table 8.15. Computation of the Next table.

8.2.3.4. KMP zoek-procedures.

Door de twee deelprogramma’s te groeperen bekomt men een zoekprocedure gelijkaardig aan deze die ontwikkeld werd voor het BF algoritme, maar nu volgens het KMP algoritme. Deze procedure vindt men in tabel 8.16.
PROCEDURE Search(VAR Text: ARRAY OF CHAR; TLength:CARDINAL;

                 VAR String: StringType; SLength:CARDINAL):CARDINAL;

   VAR j, jmax : CARDINAL;

       Next : ARRAY[1..StringSize] OF CARDINAL;

BEGIN
(* Computation of the Next table *)

     i := 1; k := 0; Next[1] := 0;

     WHILE i < SLength DO

       WHILE (k > 0) AND (String[i]#String[k]) DO k := Next[k] END;

       k := k + 1; i := i + 1;

       IF String[i] = String[k]

         THEN Next[i] := Next[k]

         ELSE Next[i] := k

       END; (* IF *)

     END; (* WHILE *)

(* Actual KMP search *)

     j := 1; i := 1;

     WHILE (i <= SLength) AND (j <= TLength) DO
       WHILE (i > 0) AND (Text[j] # String[i]) DO i := Next[i] END;

       i := i + 1; j := j + 1;

     END; (* WHILE *)

     IF i > SLength THEN RETURN j – Slength

                    ELSE RETURN Tlength

     END; (* IF *)

END Search;

Table 8.16. A straightforward implementation of the KMP search procedure.

Het is vrij duidelijk dat deze voor de hand liggende programmatie nog kan verbeterd worden. Zo zouden sentinel technieken kunnen aangewend worden om de AND expressies uit de WHILE lussen te elimineren.

Een ietwat meer subtiele verbetering is gebaseerd op de vaststelling dat, bij grote alfabetten, een verschil meestal reeds zal vastgesteld worden bij het eerste karakter van de string. Dit verschil zal dan gewoonweg tot gevolg hebben dat de string een positie voorwaarts wordt geschoven. Bij de voor de hand liggende programmatie van tabel 8.16 worden er wel vrij veel instructies uitgevoerd om de string eenvoudigweg een plaats naar voor te schuiven. Een niet verwaarloosbare winst kan geboekt worden door een verschil in het eerste karakter van de string afzonderlijk te behandelen.
De programmatie van deze verbeteringen is aan de studenten voorgesteld als oefening.

8.2.4. Het Boyer-Moore (BM) algoritme.
Het BM algoritme vertoont een zekere gelijkenis met het KMP algoritme, in die zin dat het ook gebruik maakt van tabellen die de string beschrijven om het voortschuiven van de string langs de tekst te versnellen. Het grote verschil ligt echter in de richting van de vergelijkingen: terwijl BF en KMP de string van links naar rechts doorlopen doet BM dat in de andere richting. Deze werkwijze laat toe meer informatie te verzamelen tijdens de vergelijkingen en alzo het verschuiven te versnellen.

Het BM algoritme zal hier niet verder besproken worden. De geïnteresseerde lezer wordt verwezen naar het artikel Boyer en Moore in de referenties.
8.2.5. Performanties.

Het BF algoritme gebruikt twee geneste lussen, een voor het doorlopen van de tekst en een voor het doorlopen van de string. Indien m en n respectievelijk de lengte van de tekst en van de string zijn, kunnen er tot m*n vergelijkingen gebeuren. Gelukkig ligt het eigenlijke aantal vergelijkingen meestal veel lager dan deze bovenlimiet, want verschillen worden niet steeds bij het laatste karakter van de string gevonden.
Zowel het KMP als het BM algoritme doorlopen eerst de string om zijn eigenschappen te bepalen en daarna wordt de tekst doorlopen. Beide operaties gebeuren zonder dat er ooit teruggestapt wordt. Het aantal vergelijkingen is dus van de orde van m+n in plaats van m*n. Jammer genoeg zijn de vrij complexe bewerkingen van de KMP en BM algoritmen ook tijdrovend, zodanig dat het voordeel van BM en KMP tegenover BF maar betekenisvol wordt wanneer men met zeer lange teksten en zeer lange strings te maken heeft. In deze laatste situatie is KMP vooral goed wanneer het alfabet erg beperkt is terwijl BM vooral aantrekkelijk wordt met grote alfabetten.

Appendix 8.A.  Array-search demonstration programs.

MODULE SrchDem1;

IMPORT SYSTEM;

FROM Lib IMPORT RANDOM,Dos;

FROM IO IMPORT WrLn,WrStr,WrCard,RdCard,WrLngCard;

CONST   Size =  10000;

TYPE Item = RECORD key : CARDINAL END;

VAR

    a : ARRAY [0..Size] OF Item;

    n,k,i : CARDINAL;

    Reg : SYSTEM.Registers;

    InitTime,FinalTime : LONGCARD;

(* Here come the different versions of the Search procedure *)

BEGIN

    FOR i := 1 TO Size DO

        a[i].key := i * 2

    END;

    WrStr(' Give a key between 0 and 20000 ');

    k := RdCard(); WrLn;

    Reg.AH := 2CH; Dos(Reg);

    WITH Reg DO

       InitTime := VAL(LONGCARD,CH)*360000

                 + VAL(LONGCARD,CL)*6000

                 + VAL(LONGCARD,DH)*100

                 + VAL(LONGCARD,DL);

    END;

    FOR n := 1 TO 1000 DO

      i := Search(a,k);

    END;

    Reg.AH := 2CH; Dos(Reg);

    WITH Reg DO

       FinalTime := VAL(LONGCARD,CH)*360000

                  + VAL(LONGCARD,CL)*6000

                  + VAL(LONGCARD,DH)*100

                  + VAL(LONGCARD,DL);

    END;

    IF i = 0 THEN WrStr('The key has not been found')

             ELSE WrStr('The key has been found at ');

                  WrCard(i,5);

    END;

    WrLn;

    WrStr(' This search took ');

    WrLngCard((FinalTime-InitTime)*10,16);

    WrStr(' microseconds')

END SrchDem1.

Straight search procedure.

PROCEDURE Search(VAR a: ARRAY OF Item; k:CARDINAL):CARDINAL;

   VAR i : CARDINAL;

BEGIN

   i := HIGH(a);

   WHILE (a[i].key # k) AND (i # 0) DO DEC(i) END;

   RETURN i;

END Search;

Sentinel Search procedure.

PROCEDURE Search(VAR a: ARRAY OF Item; k:CARDINAL):CARDINAL;

   VAR i : CARDINAL;

BEGIN

   i := HIGH(a);

   a[0].key := k;

   WHILE a[i].key # k DO DEC(i) END;

   RETURN i;

END Search;

Recursive binary search procedure

PROCEDURE Search(VAR a: ARRAY OF Item; k:CARDINAL):CARDINAL;

   VAR Min,Max,m : CARDINAL;

   PROCEDURE src;

   BEGIN

      m := (Min+Max) DIV 2;

      IF Min # Max  THEN

          IF a[m].key >= k

            THEN Max := m; src

            ELSE Min := m+1; src

          END;

      END;

   END src;

BEGIN

   Min := 1; Max := HIGH(a);

   Src;

   IF a[Max].key = k

      THEN RETURN Max

      ELSE RETURN 0

   END

END Search;

Iterative binary search procedure.

PROCEDURE Search(VAR a: ARRAY OF Item; k:CARDINAL):CARDINAL;

   VAR Min,Max,m : CARDINAL;

BEGIN

   Min := 1; Max := HIGH(a);

   WHILE Min < Max DO

     m := (Max+Min) DIV 2;

     IF a[m].key >= k

       THEN Max := m

       ELSE Min := m+1

     END;

   END;

   IF a[Max].key = k

      THEN RETURN Max

      ELSE RETURN 0

   END

END Search;

Appendix B.  String-search demonstration programs.

Brute-force search program:

MODULE Strsrch1;

IMPORT SYSTEM;

FROM Lib IMPORT RANDOM,RANDOMIZE,Dos;

FROM IO IMPORT WrLn,WrStr,WrCard,WrChar,RdChar,RdKey,WrLngCard;

CONST

    TextSize =  40000;

    StringSize = 20;

TYPE

    StringType = ARRAY[0..StringSize] OF CHAR;

VAR

    Text : ARRAY [0..TextSize] OF CHAR;

(* The actual text to be searched should start at position 1 *)

(* and not extend farther than position TextSize-StringSize  *)

    String : StringType;

    TextLength, StringLength : CARDINAL;

    ch1,ch2,n,k,i,j : CARDINAL;

    Reg : SYSTEM.Registers;

    InitTime,FinalTime : LONGCARD;

    Ch : CHAR;

PROCEDURE Search

           (VAR Text: ARRAY OF CHAR; TLength:CARDINAL;

            VAR String: ARRAY OF CHAR; SLength:CARDINAL):CARDINAL;

   VAR j, jmax : CARDINAL;

BEGIN

     j := 1; jmax := TLength - SLength;

     REPEAT

       WHILE (Text[j] # String[1]) AND (j <= jmax) DO j := j+1 END;

       IF j <= jmax THEN

         i := 1;

         WHILE (Text[j+i-1] = String[i]) AND (i < SLength) 

           DO i := i+1 END;

       END; (* IF *)

       j := j + 1;

     UNTIL (i = SLength) OR (j > jmax);

     RETURN j - 1

END Search;

BEGIN

(* Random text initialization *)

(* The text occupies positions 1 to TextSize-1 *)

(* three letter alphabet *)

  ch1 := ORD('a'); ch2 := ORD('c')-ch1+1;

  FOR j := 1 TO TextSize-StringSize DO

    Text[j] := CHR(ch1+RANDOM(ch2))

  END;

  TextLength := TextSize - StringSize;

  FOR j := 1 TO 79 DO WrChar(Text[j]) END; WrLn;

(* String initialization *)

  WrStr(' Type the string to be searched for, terminate with CR.');

  WrLn;

  WrStr(' Maximum number of characters = ');

  WrCard(StringSize,3); WrLn;

  StringLength := 1;

  REPEAT

   String[StringLength] := RdKey(); WrChar(String[StringLength]);

   StringLength := StringLength + 1;

  UNTIL (String[StringLength-1] = CHR(13)) 

         OR (StringLength >= StringSize);

  WrLn;

  StringLength := StringLength - 2;

(* Recording of start of search time *)

  Reg.AH := 2CH; Dos(Reg);

  WITH Reg DO

   InitTime := VAL(LONGCARD,CH)*360000

             + VAL(LONGCARD,CL)*6000

             + VAL(LONGCARD,DH)*100

             + VAL(LONGCARD,DL);

  END;

(* Perform the search 100 times *)

  FOR n := 1 TO 100 DO

   j := Search(Text, TextLength, String, StringLength);

  END;

(* Compute total duration of searches *)

  Reg.AH := 2CH; Dos(Reg);

  WITH Reg DO

   FinalTime := VAL(LONGCARD,CH)*360000

              + VAL(LONGCARD,CL)*6000

              + VAL(LONGCARD,DH)*100

              + VAL(LONGCARD,DL);

  END;

Brute-force search procedure with sentinels.

PROCEDURE Search

           (VAR Text: ARRAY OF CHAR; TLength:CARDINAL;

            VAR String: ARRAY OF CHAR; SLength:CARDINAL):CARDINAL;

 VAR j, jmax : CARDINAL;

BEGIN

 j := 1; jmax := TLength - Slength;

(* Sentinel for the scanning of Text *)

 Text[TLength+1] := String[1];

(* Sentinel for the scanning of String *)

 String[SLength+1] := CHR(0);

 REPEAT

  WHILE (Text[j] # String[1]) DO j := j+1 END;

  IF j <= jmax THEN

   i := 1;

   WHILE (Text[j+i-1] = String[i]) DO i := i+1 END;

  END; (* IF *)

  j := j + 1;

 UNTIL (i > SLength) OR (j > jmax);

 RETURN j-1

END Search;

KMP search procedure:

PROCEDURE Search

           (VAR Text: ARRAY OF CHAR; TLength:CARDINAL;

            VAR String: ARRAY OF CHAR; SLength:CARDINAL):CARDINAL;

 VAR j, jmax : CARDINAL;

        Next : ARRAY[1..StringSize] OF CARDINAL;

BEGIN

(* Computation of the Next table *)

 i := 1; k := 0; Next[1] := 0;

 WHILE i < SLength DO

  WHILE (k > 0) AND (String[i]#String[k]) DO k := Next[k] END;

  k := k + 1; i := i + 1;

  IF String[i] = String[k]

   THEN Next[i] := Next[k]

   ELSE Next[i] := k

  END; (* IF *)

 END; (* WHILE *)

(* Actual KMP search *)

 j := 1; i := 1;

 WHILE (i <= SLength) AND (j <= TLength) DO

  WHILE (i > 0) AND (Text[j] # String[i]) DO i := Next[i] END;

  i := i + 1; j := j + 1;

 END; (* WHILE *)

 IF i > SLength THEN RETURN j - SLength

                ELSE RETURN TLength

 END; (* IF *)

END Search;

